
14. CASE STUDY II: GENERAL PROBLEM SOLVER 

14.1. The requirements 

A general problem solver works around the concept of states. The object of the program is to take a starting 
state, a set of possible operations and a set of goals from the user. It attempts to achieve all the goals using the 
available operators. An example of possible operators is:  

(defparameter *school-ops*  
  (list  
   (make-op :Action 'drive-son-school  
            :preconds '(son-at-home car-works) 
            :add-list '(son-at-school) 
            :del-list '(son-at-home)) 
 
   (make-op :Action 'tell-shop-problem   
            :preconds '(in-communication-with-shop) 
            :add-list '(shop-knows-problem)))) 

In all the operators, there is an action – which identifies uniquely an operator. It has preconds – the conditions 
which have to be satisfied by a state before the operator can be applied to it. It includes an add-list and a del-list 
– these being the changes which the operator makes to the state.  

A typical call to GPS would be as follows:  

> (gps '(son-at-home  car-needs-battery   
   have-money   have-phone-book) 

       '(son-at-school) 
       *school-ops*) 

GPS would attempt to achieve the goals (the second parameter to GPS) using *school-ops* starting from the 
current state (first parameter to GPS). If it fails to achieve any of the goals, it returns NIL. Otherwise, it prints 
out the operators applied to achieve the goals.  

14.2. Analysis of GPS  

Let’s define a basic layout for achieving goals.  

First of all, to achieve a list of goals, it is necessary to achieve all elements of that list. That’s quite obvious. Let’s 
write pseudo-pseudo code for achieving a list of goals.  

1. To achieve a list of goals, achieve all goals.  
2. A goal is achieved if it’s already in the current state  
3. If it’s not already in the current state, it can be achieved using this procedure.   

a. Find an operator which is appropriate for achieving this goal. (Meaning basically that the goal 
is in the operator’s add-list. When the operator will be applied, the goal will be added to the 
state – thus achieved.)  

b. To apply this appropriate operator, we need to achieve all the goals in the preconds of the 
operator. Only then can that operator be applied.  

c. But we already know how to achieve a goal. Recursive call to achieve goal here!  

All we have to do now is to translate this code to lisp code. This may not be so intuitive to all of us. So, here’s the 
code.  

14.3. Helper functions 

find-all: the objective of this function is to take a list and a function. It removes all elements of the list 
which do not satisfy the function (usually a predicate) thus leaving only the elements which we want to find. A 
sample call to find-all is given before the code.  

> (find-all 1 ‘(1 2 3 4 1) :test #’= :key #’abs) ⇒ (1 1) 
 
(defun find-all  (item sequence &rest keyword-args  
                       &key (test #'eql) test-not &allow-other-keys) 
  "Find all those elements of sequence that match item 
   according to the keywords. Doesn't alter sequence"  
 
  (if test-not  
      (apply #'remove item sequence  
             :test-not (complement test-not) keyword-args) 
    (apply #'remove item sequence  
           :test (complement test) keyword-args))) 

Look up the description of complement in a reference. It takes a function and returns another function. The 
new returned function is like a not operator for the original function. If the original function was going to 
return t on some operands, this new function would return nil and vice versa.  

We pass appropriate-p as the test to find-all in our GPS. It makes sure that we get only the ops as a 
result which are appropriate for a goal. Here’s the listing for appropriate-p.  

(defun appropriate-p (goal op) 
  "An op is appropriate to a goal if it is in it’s add list" 
  (member goal (op-add-list op))) 

14.4. The core of the code  

Here’s the complete code for our simple GPS.  

(defvar *state* nil "The current state: a list of conditions") 
(defvar *ops* nil "A list of operators available") 
 
(defstruct op "An operation" 
  (action nil)  
  (preconds nil)  
  (add-list nil)  
  (del-list nil)) 
(defun GPS (*state* goals *ops*) 
  "General problem solver: achive all goals using *ops*" 
  (if (every #'achieve goals) 'solved)) 
 
(defun achieve (goal) 
  "A goal is achieved if it already holds  
   or if there is an op for it that is applicable" 
  (or (member goal *state*) 
      (some #'apply-op 
            (find-all goal *ops* :test #'appropriate-p)))) 

(defun appropriate-p (goal op) 
  "An op is appropriate to a goal if it is in it's add list" 
  (member goal (op-add-list op))) 
 
(defun apply-op (op) 
  "Prints a message and update *state* if op is applicable" 
  (when (every #'achieve (op-preconds op)) 
    (print (list 'executing (op-action op))) 
    (setf *state* (set-difference *state* (op-del-list op))) 
    (setf *state* (union *state* (op-add-list op))) 
    t)) 

Notice we call achieve from two places – one at the beginning and the second from within achieving another 
goal. Take a look at the analysis if you need to see where this comes from.  

To use the GPS, user has to define the ops:  

(defparameter *school-ops*  
  (list  
   (make-op :Action 'drive-son-school  
            :preconds '(son-at-home car-works) 
            :add-list '(son-at-school) 
            :del-list '(son-at-home)) 
   (make-op :action 'shop-installs-battery  
            :preconds '(car-needs-battery shop-knows-problem shop-has-
money) 
            :add-list '(car-works)) 
   (make-op :Action 'tell-shop-problem   
            :preconds '(in-communication-with-shop) 
            :add-list '(shop-knows-problem)) 
   (make-op :Action 'telephone-shop 
            :preconds '(know-phone-number) 
            :add-list '(in-communication-with-shop)) 
   (make-op :Action 'look-up-number 
            :preconds '(have-phone-book) 
            :add-list '(know-phone-number)) 
   (make-op :Action 'give-shop-money 
            :preconds '(have-money) 
            :add-list '(shop-has-money) 
            :del-list '(have-money)))) 

And call GPS with a starting state and a list of goals.  

(gps '(son-at-home  
       car-needs-battery  
       have-money  
       have-phone-book) 
     '(son-at-school) 
     *school-ops*) 

Practice a little with the calls and you’ll see that this GPS is not a very bug-free program. A complete discussion 
of this can be seen in [NORVIG2]. I’ll trust you to find some problems on your own.  

 

 


